```
pro driver, input, output, seed=seed

;;*****************************************************************************
;;
;;  Driver routine to run the 5th order RK integrator from Numerical
;;  Recipies, 3rd Ed.
;;
;;  Version History:
;;   3.1: 4/27/2011
;;     -- Need to speed up when modeling satellites
;;   3.0: 7/20/2010
;;     -- Revising for new structure architecture
;;   2.7: 7/6/2010:
;;     -- Adding impact_check_2.9 to this program so it does not use the include
;;   2.6: 4/26/2010:
;;     -- Added moon's temperature map
;;     -- removed thermalized option from emitfn case (no longer used)
;;   2.5: 1/14/2010:
;;     -- Keep track of fate of each packet
;;     -- Keep track of deposition on the surface
;;     -- Replace ptr_free with destory_structure
;;   2.4: 12/7/2009
;;     -- Allowing variable surface temperature for Maxwellian reemission
;;   2.3: 11/6/2009
;;     -- changing the way it does the thermalization. New velocity is determined from
;;         partial accomodation to thermal speed at surface
;;   2.2: Added variable surface temperature for particle sticking (for Mercury,
;;     Based on surface temperature in Leblanc & Johnson 2003). -- I don't think I
;;     did this [12/7/09]
;;   2.1: Added support for elastic bouncing of particles from the surface
;;   2.0: Revised for current structure setup
;;   1.2: Previous working version
;;   1.1: Older version to work with rk4
;;   1.0: Similar to the original version to work with rk7
;;
;; Impact_check version history (before inclusion into this program):
;;   Version 2.9 4/28/10
;;      -- Changing definition of accommodation coefficient
;;         * Need energy accommodation rather than velocity accommodation
;;         * Before: v_1 = a v_th + (1-a) v_0
;;         * After: v_1^2 = a v_th^2 + (1-a) v_0^2
;;   Version 2.8 3/9/10
;;      -- Fixing issue with thermal accomodation - Now choose speed based on
;;     thermal distribution.
;;   Version 2.7 1/19/10
;;      -- fixing some problems with v2.6
;;   Version 2.6 1/14/2010
;;      -- keep track of what happens to each packet
;;      -- keep track of surface deposition
;;           * doesn't get the map right for satellites.
;;   Version 2.4 11/6/2009
```

```
;;      -- added thermal accomodation to the surface
;;   Version 2.1 -- added option for elastic bouncing
;;
;;*****************************************************************

common constants

;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;
;; Remake the loc structure to speed up the math and make it easier to read
if (input.options.trackloss) $
  then loc = {t:ptr_new(0), x:ptr_new(0), v:ptr_new(0), frac:ptr_new(0), $
    lossfrac:ptr_new(0), hitfrac:ptr_new(0), ringfrac:ptr_new(0)} $
  else loc = {t:ptr_new(0), x:ptr_new(0), v:ptr_new(0), frac:ptr_new(0)}
*loc.x = [[*output.x], [*output.y], [*output.z]]
*loc.v = [[*output.vx], [*output.vy], [*output.vz]]
*loc.frac = *output.frac
*loc.t = *output.time ;; This is how much time before present and works up to zero
npack = n_elements(*output.x)

if (input.options.trackloss) then begin
  lossfrac = dblarr(npack)
  hitfrac = dblarr(npack,n_elements(*SystemConsts.objects))
  ringfrac = dblarr(npack)
  leftfrac = dblarr(npack)
  deposition = {longitude:ptr_new(findgen(360)*!dtor), $
    latitude:ptr_new(findgen(180)*!dtor-!pi/2), $
    map:ptr_new(dblarr(360,180,n_elements(*SystemConsts.objects)))}
  ilon = findgen(360)
  ilat = findgen(180)
endif

which = where(*input.geometry.include, nw)
pp = replicate(1., n_elements(*SystemConsts.objects))

;; Set up the stepsizes
h = replicate(1000d, npack)     ;initial guess at best stepsize
hold = h  ;; last step used by each packet

;Set variables in preparation for iteration
count = 0L  ;; number of steps taken

;These control how quickly the stepsize is increased or decreased between iterations
safety = .95
shrink = -.25
grow = -.2

;; yscale = scaling parameter for each variable
;;   x,y,z ~ R_plan
;;   vx,vy,vz ~ 1 km/s (1/Rplan Rplan/s)
;;   frac ~ exp(-t/lifetime) ~ mean(frac)

```

```
103  resolution = input.options.resolution
104
105  ;; Set up the bounce conditions
106  if (input.sticking_info.stickcoef NE 1.) then begin
107    case strlowcase(input.sticking_info.emitfn) of
108      'maxwellian': begin
109        if (input.sticking_info.Tsurf EQ 0) then begin   ;; Use surf temp model
110          case (input.geometry.startpoint) of
111            'Mercury': begin
112              temp0 = 100.
113              temp1 = 600 + 125*(cos(input.geometry.taa)-1)/2. ;; sub-solar temp fn of taa
114              nn = .25
115              end
116            'Moon': begin
117              temp0 = 151
118              temp1 = 162.
119              nn = .25
120              end
121            else: stop
122          endcase
123
124          nt = 21 & nv = 101
125          temperature = dindgen(nt)/(nt-1)*temp1 + temp0
126          v_temp = sqrt(2*temperature*!const.kb/atomicmass(input.options.atom)) /1e5
127          prob = dindgen(101)/100.
128          vgrid = dblarr(nt,101)
129          for i=0,nt-1 do begin
130            vrange = dindgen(nv)/(nv-1)*v_temp[i]*3.
131            f_v = MaxwellianDist(vrange, temperature[i], input.options.atom)
132            sumdist = f_v
133            for j=1,nv-1 do sumdist[j] += sumdist[j-1]
134            sumdist /= max(sumdist)
135            vgrid[i,*] = interpol(vrange, sumdist, prob)
136          endfor
137        endif else begin   ;; use constant surf temp
138          v_th = sqrt(2*input.Sticking_info.Tsurf*!const.kb/atomicmass(input.options.atom)) /1e5
139          vrange = findgen(1001)/1000 * v_th*5 & vrange = vrange[1:*]
140          f_v = MaxwellianDist(vrange, input.Sticking_info.Tsurf, options.atom)
141
142          sumdist = f_v
143          for i=1,n_elements(vrange)-1 do sumdist[i] += sumdist[i-1]
144          sumdist /= max(sumdist)
145        endelse
146        end
147      'elastic scattering':
148      else: stop
149    endcase
150  end
151
152  ;*************************************************************************
153  ;Keep takeing R.K. steps until every packet has reached the time of "image taken"
```

```
154  ;*************************************************************************
155
156  moretogo = where((*loc.t GT resolution) and (*loc.frac GT 0), ntogo)
157  done = (ntogo EQ 0)
158
159  while ~(done) do begin
160    ;Now generate sub-arrays containing only the particles that are still being tracked
161
162    loc0 = {t: ptr_new((*loc.t)[moretogo]), x:ptr_new((*loc.x)[moretogo,*]), $
163      v:ptr_new((*loc.v)[moretogo,*]), frac:ptr_new((*loc.frac)[moretogo])}
164
165    w = where(*loc0.frac EQ 0, nw) & if (nw NE 0) then stop
166    w = where(finite(*loc0.x) EQ 0, nw) & if (nw NE 0) then stop
167    w = where(finite(*loc0.v) EQ 0, nw) & if (nw NE 0) then stop
168
169    oldx = *loc0.x  ;; This is used for determining if anything hit the rings
170    oldf = *loc0.frac
171    h = hold[moretogo]
172
173    ;Adjust stepsize to be no more than time remaining
174    h = (h LE (*loc0.t))*h + (h GT (*loc0.t))*(*loc0.t)
175
176    ;; Run the rk5 step
177    rk_5, loc0, h, input, which, delta
178
179    ;; Do the error check
180    ;; scale = a_tol + |y| * r_tol
181    ;;   for x: a_tol = r_tol = resolution
182    ;;   for v: a_tol = r_tol = resoltuon/10. -- require v to be more precise
183    ;;   for f: a_tol = 0.01 ; r_tol = 0 -- set fractional tolerance to 1%
184    scalespace = resolution + abs(*loc0.x) * resolution
185    scalevel = 0.1*(resolution + abs(*loc0.v) * resolution)
186    scaleabund = 0.01 ;; resolution + abs(*loc0.frac) * resolution
187
188    ;; difference relative to acceptable difference
189    *delta.x /= scalespace
190    *delta.v /= scalevel
191    *delta.frac /= scaleabund
192    xerrmax = max(*delta.x, dim=2)
193    verrmax = max(*delta.v, dim=2)
194
195    ;; Maximum error for each packet
196    errmax = (xerrmax GE verrmax)*xerrmax + (xerrmax LT verrmax)*verrmax
197  ;  errmax = (errmax GE *delta.frac)*errmax + (errmax LT *delta.frac)**delta.frac
198    if ((where(finite(errmax) EQ 0))[0] NE -1) then stop
199
200    ;; Check where difference is very small - adjust step size
201    noerr = where(errmax LE 1e-7)
202    if (noerr[0] NE -1) then begin
203      errmax[noerr] = 1.
204      h[noerr] = h[noerr]*10.
```

```
205     endif
206
207     ;; Put the post-step values
208     g = where(errmax LE 1.0, ng, comp=b)
209     if (ng GT 0) then begin
210       (*loc.t)[moretogo[g]] = (*loc0.t)[g]
211       (*loc.x)[moretogo[g],*] = (*loc0.x)[g,*]
212       (*loc.v)[moretogo[g],*] = (*loc0.v)[g,*]
213       (*loc.frac)[moretogo[g]] = (*loc0.frac)[g]
214       if (input.options.trackloss) then $
215         lossfrac[moretogo[g]] += (oldf[g]-(*loc0.frac)[g])  ;; add in change in frac
216       h[g] = safety*h[g]*errmax[g]^grow
217     endif
218
219     if (ng NE ntogo) then begin
220       ;; don't adjust the bad values, but do fix the stepsize
221       htemp = safety * h[b] * errmax[b]^shrink
222       q = where(htemp LT 0.0, nq) & if (nq NE 0) then stop
223
224       ;; don't let step size drop below 1/10th previous step size
225       h[b] = max([[htemp], [0.1*h[b]]], dim=2)
226     endif
227     qqq = where(h LT 1e-7, nqq) & if (nqq NE 0) then stop  ;; error test
228
229     destroy_structure, loc0
230     destroy_structure, delta
231
232     ;save new values of h
233     hold[moretogo] = h
234
235     ;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;
236     ;; Impact check
237     ;; Only look at packets which moved during this step
238     if (ng GT 0) then begin
239       ;; Make a new structure with just the packets that moved this step
240       loc1 = {t:ptr_new((*loc.t)[moretogo[g]]), x:ptr_new((*loc.x)[moretogo[g],*]), $
241         v:ptr_new((*loc.v)[moretogo[g],*]), frac:ptr_new((*loc.frac)[moretogo[g],*])}
242       if (input.options.trackloss) then begin
243         oldfrac = *loc1.frac
244         hitfrac1 = hitfrac[moretogo[g],*]
245         ringfrac1 = ringfrac[moretogo[g]]
246         leftfrac1 = leftfrac[moretogo[g]]
247       endif
248
249       jj = replicate(1., ng)
250
251       ;; 1) Did the packets hit anything?
252       ;Get object positions
253       if (input.options.motion) $
254         then locmoon, *loc1.t, *input.geometry.phi, *SystemConsts.a, $
255           *SystemConsts.orbrate, x=xSat, y=ySat, z=zSat, ang=ang $
```

```
256         else locmoon, fltarr(ng), *input.geometry.phi, *SystemConsts.a, $
257           *SystemConsts.orbrate, x=xSat, y=ySat, z=zsat, ang=ang
258
259       ;; Distance of packets from each object
260       tempR = sqrt(((*loc1.x)[*,0]#pp - xSat)^2 + ((*loc1.x)[*,1]#pp - ySat)^2 + $
261         ((*loc1.x)[*,2]#pp - zSat)^2)
262
263       ;Is r < satellite radius?
264       eps = 0.
265       satrad = jj # (*SystemConsts.radius)*(1-eps)
266       hhh = where((tempR-satrad) LT 0, nhits)
267       if (nhits NE 0) then begin
268         ;;w = where((*loc1.t)[hhh mod ng] EQ 0, nw) & if (nw NE 0) then stop
269         hx = hhh mod ng & hy = hhh/ng
270
271         ;; adject the frac values
272         if (input.sticking_info.stickcoef EQ 1) $
273           then (*loc1.frac)[hx] = 0 $
274           else (*loc1.frac)[hx] = (*loc1.frac)[hx] * (1.-input.sticking_info.stickcoef)
275
276         ;; If need to know where things hit the surface, do this
277         if ((input.options.trackloss) or (input.sticking_info.stickcoef LT 1)) then begin
278           ;; Figure out where things hit the surface
279           srad = satrad[hhh]
280           r0 = tempR[hhh]  ;; R_plan
281           x0 = (*loc1.x)[hx,0]  ;; R_plan
282           y0 = (*loc1.x)[hx,1]  ;; R_plan
283           z0 = (*loc1.x)[hx,2]  ;; R_plan
284           r0 = sqrt(x0^2 + y0^2 + z0^2)
285
286           ;; Position of the satellites
287           xcent = xSat[hx,hy]
288           ycent = ySat[hx,hy]
289           zcent = zSat[hx,hy]
290
291           ;; Vector from center of satellite to packet
292           ;;   -- packet positions relative to satellite
293           x1 = (x0-xcent)/srad  ;; rsat
294           y1 = (y0-ycent)/srad  ;; rsat
295           z1 = (z0-zcent)/srad  ;; rsat
296
297           ;; Velocity - orbital vel = vel relative to satellite
298           vxsat = -(*SystemConsts.orbvel)[hy]*cos(ang[hx,hy])*input.options.motion/$
299             SystemConsts.rplan
300           vysat = -(*SystemConsts.orbvel)[hy]*sin(ang[hx,hy])*input.options.motion/$
301             SystemConsts.rplan
302
303           vx0 = (*loc1.v)[hx,0] - vxsat  ;; rplan/s
304           vy0 = (*loc1.v)[hx,1] - vysat  ;; rplan/s
305           vz0 = (*loc1.v)[hx,2]  ;; rplan/s
306
```

```
307           ;; Find where the packet hit the surface
308           ;;    |x + vt| = 1 -- see ResearchNotes from 4/28/08
309           a = vx0^2 + vy0^2 + vz0^2
310           b = 2*(x1*vx0 + y1*vy0 + z1*vz0)
311           c = x1^2 + y1^2 + z1^2 - 1
312
313           dd = b^2 - 4*a*c
314           q = where(dd LT 0, nq) & if (nq NE 0) then stop
315           t0 = (-b - sqrt(b^2-4*a*c))/(2*a)
316           t1 = (-b + sqrt(b^2-4*a*c))/(2*a)
317           t = (t0 LE 0)*t0 + (t1 LT 0)*t1
318
319           ;; Point where packet hit the surface
320           x2 = x1 + vx0*t
321           y2 = y1 + vy0*t
322           z2 = z1 + vz0*t
323           ;; r2 = sqrt(x2^2 + y2^2 + z2^2)  ;; -- this should be = 1.
324
325           lonhit = (atan(x2, -y2) + 2*!pi) mod (2*!pi)
326           lathit = asin(z2)
327
328           ;; Put new coordinates into the array
329           x_final = xcent + x2*srad
330           y_final = ycent + y2*srad
331           z_final = zcent + z2*srad
332
333           q = where(finite(x_final) EQ 0, nq) & if (nq NE 0) then stop
334           q = where(finite(y_final) EQ 0, nq) & if (nq NE 0) then stop
335           q = where(finite(z_final) EQ 0, nq) & if (nq NE 0) then stop
336           (*loc1.x)[hx,0] = x_final
337           (*loc1.x)[hx,1] = y_final
338           (*loc1.x)[hx,2] = z_final
339         endif
340
341         if (input.options.trackloss) then begin
342           lonind = (fix(interpol(ilon, *deposition.longitude, lonhit))) mod $
343             n_elements(*deposition.longitude)
344           latind = fix(interpol(ilat, *deposition.latitude, lathit)) mod $
345             n_elements(*deposition.latitude)
346           for i=0L,nhits-1 do (*deposition.map)[lonind[i],latind[i],hy[i]] += $
347             oldfrac[hx[i]]*input.sticking_info.stickcoef
348           hitfrac1[hhh] += oldfrac[hx]
349         endif
350
351         if (input.sticking_info.stickcoef LT 1) then begin
352           ;; Figure out rebound velocity
353           vv02 = vx0^2 + vy0^2 + vz0^2  ;; rplan/s
354           PE = 2*(*SystemConsts.GM)[hy]*(1./r0-1./srad)
355           vv02 += PE
356           q = where(vv02 LT 0, nq) & if (nq NE 0) then vv02[q] = 0.
357           q = where(finite(vv02) EQ 0, nq) & if (nq NE 0) then stop
```

```
358
359           case strlowcase(input.sticking_info.emitfn) of
360             'maxwellian': begin ;; Re-emit the packets with a thermal distribution
361               if (input.sticking_info.Tsurf EQ 0) then begin
362                 surftemp = temp0 + (temp1*(abs(cos(lonhit)*cos(lathit)))^nn)*$
363                   (abs(lonhit) LT !pi/2)
364                 rr = random_nr(nhits, seed=seed, routine=0)
365                 vv_new = interpolate_xy(vgrid, temperature, prob, surftemp, rr)/$
366                   SystemConsts.rplan
367               endif else vv_new = interpol(vrange, sumdist, $
368                 random_nr(seed=seed, nhits))/SystemConsts.rplan ;; rplan/s
369
370               vv2 = sqrt(input.Sticking_info.accom_factor*vv_new^2 + $
371                 (1-input.Sticking_info.accom_factor)*vv02)
372               end
373             'elastic scattering': vv2 = sqrt(vv02)
374           endcase
375
376           ;; Determine new direction with F(v) ~ cos(theta)
377           alt = acos(random_nr(seed=seed, nhits))
378           az = 2*!pi * random_nr(seed=seed, nhits)
379
380           v_rad = sin(alt)          ;; Radial component of velocity
381           v_east = -cos(alt) * sin(az) ;; Component along latitude line (points east)
382           v_north = cos(alt) * cos(az) ;; Component along longitude line (points to NP)
383
384           vx2 = v_rad*x2
385           vy2 = v_rad*y2
386           vz2 = v_rad*z2
387
388           lat = asin(z2)
389           lon = atan(x2, y2)
390
391           vx2 = dblarr(nhits) & vy2 = dblarr(nhits) & vz2 = dblarr(nhits)
392           for i=0L,nhits-1 do begin
393             M = transpose([ $
394               [cos(lat[i])*sin(lon[i]), cos(lat[i])*cos(lon[i]), sin(lat[i])], $
395               [-cos(lon[i]), sin(lon[i]), 0], $
396               [-sin(lat[i])*sin(lon[i]), -sin(lat[i])*cos(lon[i]), cos(lat[i])] ])
397             v_ren = [v_rad[i], v_east[i], v_north[i]]
398             v_xyz = invert(M) # v_ren
399             vx2[i] = v_xyz[0] * vv2[i]
400             vy2[i] = v_xyz[1] * vv2[i]
401             vz2[i] = v_xyz[2] * vv2[i]
402           endfor
403
404           ;; The new position in planet-centered coords
405           vx_final = vx2 + vxsat
406           vy_final = vy2 + vysat
407           vz_final = vz2
408
```

```
409           q = where(finite(vx_final) EQ 0, nq) & if (nq NE 0) then stop
410           q = where(finite(vy_final) EQ 0, nq) & if (nq NE 0) then stop
411           q = where(finite(vz_final) EQ 0, nq) & if (nq NE 0) then stop
412
413           (*loc1.v)[hx,0] = vx_final
414           (*loc1.v)[hx,1] = vy_final
415           (*loc1.v)[hx,2] = vz_final
416         endif
417       endif
418
419       ;; 2) Have the packets left the corona? (only check if not tracking the full system)
420       if ~(input.options.fullSystem) then begin
421         leftCor = where(tempR[*,stuff.s] GT input.options.OuterEdge * $
422           (*SystemConsts.radius)[stuff.s], hh)
423         if (hh NE 0) then begin
424           if (input.options.trackloss) then leftfrac1[leftcor] += (*loc1.frac)[leftcor]
425           (*loc1.frac)[leftCor] = 0
426         endif
427       endif
428
429       ;; 3) If Saturn, check to see if anything hit the rings
430       if (input.geometry.planet EQ 'Saturn') then begin
431         ;; Ring region within 2.3 Rs
432         ox = oldx[g,*]
433         cross = ox[*,2] * (*loc1.x)[*,2]  ;; if cross is negative, then crossed eq. plane
434         MayHit = where(cross LE 0 , nmay)
435         if (nmay NE 0) then begin
436           orho = sqrt(total(ox[MayHit,0:1]^2, 2))
437           nrho = sqrt(total((*loc1.x)[MayHit,0:1]^2, 2))
438           w = where((orho LT 2.3) or (nrho LT 2.3), nw)
439           for j=0,nw-1 do begin $
440             crosspt = interpol([orho[w[j]],nrho[w[j]]], [ox[MayHit[w[j]],2], $
441               (*loc1.x)[MayHit[w[j]],2]], 0.)
442             if (crosspt LT 2.3) then begin
443               if (input.options.trackloss) then ringfrac1[MayHit[w[j]]] += $
444                 (*loc1.frac)[MayHit[w[j]]]
445               (*loc1.frac)[MayHit[w[j]]] = 0.
446             endif
447           endfor
448         endif
449       endif
450
451       ;; Check to see if any packets have shrunk out of existence
452       q = where((*loc1.frac GT 0) and (*loc1.frac LT 1e-10), nq)
453       if (nq NE 0) then (*loc1.frac)[q] = 0.
454
455       ;; If any new hits, set the time remaining to 0.
456       w = where(*loc1.frac EQ 0, nw)
457       if (nw NE 0) then (*loc1.t)[w] = 0.
458
459       ;Put new values back into original array (again)
```

```
460       (*loc.t)[moretogo[g]] = *loc1.t
461       (*loc.x)[moretogo[g],*] = *loc1.x
462       (*loc.v)[moretogo[g],*] = *loc1.v
463       (*loc.frac)[moretogo[g]] = *loc1.frac
464       if (input.options.trackloss) then begin
465         hitfrac[moretogo[g],*] = hitfrac1
466         ringfrac[moretogo[g]] = ringfrac1
467         leftfrac[moretogo[g]] = leftfrac1
468       endif
469
470       destroy_structure, loc1
471     endif
472     ;;end impact check
473     ;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;
474
475     moretogo = where(*loc.t GT resolution, ntogo)  ;; check to see which ones aren't done
476     if (count mod 100 EQ 0) then print, stuff.strstart + 'Step Number: ' + string(count) + $
477       ', Packets Remaining: ' + string(ntogo)
478     count += 1 ;; step counter
479
480     ;If it goes 100000 steps then it will never stop!
481     done = ((ntogo EQ 0) or (count GT 100000.))
482   endwhile
483
484   *output.x = reform((*loc.x)[*,0])
485   *output.y = reform((*loc.x)[*,1])
486   *output.z = reform((*loc.x)[*,2])
487   *output.vx = reform((*loc.v)[*,0])
488   *output.vy = reform((*loc.v)[*,1])
489   *output.vz = reform((*loc.v)[*,2])
490   *output.frac = *loc.frac
491   if (input.options.trackloss) then begin
492     *output.hitfrac = reform(hitfrac)
493     *output.lossfrac = lossfrac
494     *output.ringfrac = ringfrac
495     *output.leftfrac = leftfrac
496     output.deposition = deposition
497   endif
498
499   destroy_structure, loc
500
501   end
502
```