```
function produce_image, files, savefile=savefile

common constants
common results

;;;;;;;;;;;;;;;;
;; Determine the image origin
s = (where(strcmp(*SystemConsts.objects, format.geometry.origin, /fold), ns))[0]
if (ns NE 1) then stop

;;;;;;;;;;;;;;;;
;; Determine image field of view and rotation
geometry = format.geometry

image = dblarr((geometry.dims)[0],(geometry.dims)[1])
immin = geometry.center - geometry.width/2.
immax = geometry.center + geometry.width/2.

scale = geometry.width/(geometry.dims-1)   ;; [xscale,zscale] in Rplan/pix
Apix = (scale[0]*scale[1])*((*SystemConsts.radius)[s]*SystemConsts.rplan*1e5)^2
   ;; cm^2/pix

;; xaxis and zaxis in Robj measured from center of object
xaxis = findgen((geometry.dims)[0])*scale[0] + immin[0]
zaxis = findgen((geometry.dims)[1])*scale[1] + immin[1]

;; Determine frame rotation
M = determine_image_rotation(input, format)

;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;;
for ff=0,n_elements(files)-1 do begin
  ;; restore output file and extract useful packets
  ;; pts_sun is in solar reference frame with origin=Object center, units R_obj
  ;; vels_sun in km/s
  results_loadfile, files[ff], pts_sun, vels_sun, frac, /keepall
  radvel_sun = vels_sun[*,1] + stuff.vrplanet  ;; for g-value
     ;; note -- want to keep the ones with frac = 0 to make sure those regions
     ;; are counted as not contributing

  ;; Rotate the packets to observer frame
  pts_obs = M ## pts_sun    ;; observer along -y axis
  vels_obs = M ## vels_sun

  ;; Determine which packets are not blocked by the planet
  rhosqr_obs = pts_obs[*,0]^2 + pts_obs[*,2]^2  ;; rho in observer's frame
  inview = ((rhosqr_obs GT 1) or (pts_obs[*,1] LT 0))
  frac *= inview

  rhosqr_sun = pts_sun[*,0]^2 + pts_sun[*,2]^2
  out_of_shadow = ((rhosqr_sun GT 1) or (pts_sun[*,1] LT 0))

```

```
52    ;; Determine which packets are in the FOV
53    h = where((pts_obs[*,0] GE immin[0]) and (pts_obs[*,0] LE immax[0]) and $
54      (pts_obs[*,2] GE immin[1]) and (pts_obs[*,2] LE immax[1]), nh)
55    if (nh GT 0) then begin
56      out = {x:ptr_new(pts_obs[h,0]), y:ptr_new(pts_obs[h,1]), z:ptr_new(pts_obs[h,2]), $
57        frac:ptr_new(frac[h]), radvel_sun:ptr_new(radvel_sun[h])}
58
59      ;; Packet weighting
60      weight = results_packet_weighting(out, out_of_shadow)
61
62      ;; Additional factors:
63      case (format.quantity) of
64        'column': weight /= Apix
65        'intensity': weight /= Apix
66        'density': weight /= Vpix
67        else: stop
68      endcase
69
70      ;; Now make the image
71      newh = where(weight GT 0, nh)
72      if (nh GT 0) then begin
73        qx = round(interpol(findgen((geometry.dims)[0]), xaxis, (*out.x)[newh]))
74        qz = round(interpol(findgen((geometry.dims)[1]), zaxis, (*out.z)[newh]))
75        for j=0,nh-1 do image[qx[j],qz[j]] += weight[newh[j]]
76      endif
77      ;tv, bytscl(image)
78    endif ;; (nh GT 0)
79    print, 'Completed image ' + strint(ff) + ' of ' + strint(n_elements(files))
80  endfor
81
82  result = {image:ptr_new(image), xaxis:ptr_new(xaxis), zaxis:ptr_new(zaxis), $
83    format:format}
84  return, result
85
86  end
```